#-------------------------------------------------
#
# Project created by QtCreator 2011-11-17T09:37:12
#
#-------------------------------------------------
QT += core gui script scripttools xml sql network
CONFIG += uitools
TARGET = uiscripter
win32 { DESTDIR = build }

TEMPLATE = app

SOURCES += main.cpp

RESOURCES += \
    resources.qrc

CONFIG(debug, debug|release) {
    CONFIG += console
}

# If mobile device add mobility feature
unix:android-g++ {
    message(android build mobility)
    CONFIG += mobility
    DEFINES += Q_OS_ANDROID
    DEFINES += QT_NO_SCRIPTTOOLS
}

contains(CONFIG, mobility) {
    MOBILITY = location
    DEFINES += Q_MOBILITY

}

include(network/network.pri)
include(storage/storage.pri)
include(core/core.pri)
include(script/script.pri)
include(database/database.pri)
include(gui/gui.pri)
include(zip/zip.pri)

# Directories for generated files
unix {
    android-g++ {
        OBJECTS_DIR = build/.obj/android
        MOC_DIR = build/.moc/android
        RCC_DIR = build/.rcc/android
        UI_DIR = build/.ui/android        
    } else {
        OBJECTS_DIR = build/.obj/linux
        MOC_DIR = build/.moc/linux
        RCC_DIR = build/.rcc/linux
        UI_DIR = build/.ui/linux
    }
}
win32 {
    OBJECTS_DIR = build/.obj/win32
    MOC_DIR = build/.moc/win32
    RCC_DIR = build/.rcc/win32
    UI_DIR = build/.ui/win32
}
























